

1. 系统概述

在包装工业中，贴标是将一件产品送入市场的整个过程中的一步。这篇 MotionCode 提供的实例集中讨论了一台贴标机（如图 1），这台贴标机可将商品的商标贴在纸板箱上。标签覆盖了整个纸箱的长度，所以标签必须与箱子的前沿对齐。

过程的开始是箱子在传送带上以一个不变的速度向贴标机进给。机械上的固定装置将箱子之间分开一个固定的距离，并推动箱子沿传送带的方向前进。

贴标机的机械系统包括一个驱动轮，一个贴标轮，和一个卷轴。驱动轮间歇性地拖动标签带运动，标签带从卷轴中被拉出，同时经过贴标轮，贴标轮会将标签带压在箱子上。在卷轴上采用了开环的位移控制，用来保持标签带的张力。

因为标签在标签带上是彼此紧密相连的，所以标签带必须不断起停。标签是在贴标轮与箱子移动速度相同的情况下被贴在箱子上的。当传送带到达了某个特定的位置时，标签带驱动轮会加速到与传送带匹配的速度，贴上标签后，再减速到停止。

由于标签带有可能产生滑动，所以它上面有登记标志，用来保证每一张标签都被正确地放置。登记标志通过一个传感器来读取，在标签带减速阶段，驱动轮会重新调整位置以修正标签带上的任何位置错误。




图 1. 贴标机

2. 要求

本段总结了对图1中贴标控制系统的一些要求。

- 用来捕捉晶片边沿位置的高速位置锁存器三个轴：传送带（A），贴标驱动轮（B），和贴标带卷轴（C）
 - 零点输入，用来帮助传送带定位零点
 - 标签带驱动轮对传送带的区域电子齿轮同步
 - 通过高速位置锁存输入信号读取登记标志位置信息
- (5). 标签带驱动轮可根据登记标志实时地修正位置
- (6). 不需要用户界面

3. 部件的选择

此段描述了为实现晶片定位机控制系统，您可以选择的 GALIL 的硬件及软件产品。下面是一个主要部件的列表，后面附有简单描述。

表 1. 贴标机控制系统需要的材料清单

部件名称	描述	价格 (U.S.) 单买 / 批量
DMC-2133	带有以太网及 RS-232 单轴控制器及驱动系统	\$1045/\$725
-DIN	DIN轨道固定选件	\$100/\$50
-DC24	直-直变频器选件，18V到36V	\$100/\$70
AMP-20540	带有 4 个 500W PWM 有刷或无刷电机驱动器的放大器	\$795/ \$495
BLM-N23-50-1000 或同等产品	带有 100 线编码器及 Hall 传感器的 Nema 23 无刷伺服电机	请咨询生产商
CPS-12-24 或同等产品	24V，200W 电源	请咨询生产商
WSDK_Servo_Tuning Software	伺服调试及分析软件	

控制器：DMC-2133-DC24-DIN

因为贴标机不许可要主机，因此我们选择 DMC-2133 独立式运动控制器。可以用一台笔记本电脑通过以太网线或者 RS-232 与控制器连接，用 WSDK 软件对伺服进行调试以及下载应用程序等操作。



图 2. CDMC-21x3. 1 到 4 轴（上）和 5 到 8 轴（下）

电机：BLM-N23-50-1000

为了达到免维护运行，我们选择无刷电机。由于晶片定位机需要不到 0.3Nm 的连续转矩，所以我们推荐使用 Galil 的 NEMA 23 #BLM-N23-50-1000 无刷电机或者其他同等电机。

电机上装有一台每转 1000 脉冲的增量编码器，积分后每转产生 4000 个脉冲。由于增量编码器为输入信号提供了通向放大器的交换通道，电机上不许可要安装 Hall 传感器。



放大器：AMP-20540

我们选择简洁紧凑的 AMP-20540 来驱动电机，该放大器可驱动 4 个轴的有刷或无刷电机，每个轴的驱动能力为 500W。该放大单元直接与 DMC-2133 连接，省去了一切通常情况下控制器与放大器之间连接所需要的连线。

图 3. AMP-20540 四轴无刷伺服放大器（每轴 500W）

4. 功能实现

此段详细介绍了如何通过上述已选择的部件来实现控制系统。

电子齿轮(GA,GR,GD)

Galil 的电子齿轮功能用来在传送带（主轴）和标签带驱动轮（辅轴）之间建立一种比例关系。

GA 命令用来设置哪个轴为主动轴，哪个轴为从动轴，**GR** 命令用来设置齿轮比。**GD** 命令用来设置从动轴与主动轴啮合时主动轴走过距离。

GAB = A ; 设置传送带为电子齿轮的主动轴
GDB = 800 ; 标签带驱动轮与主动轴啮合距离为传送带侧的 800 个脉冲
GRB = 0.75 ; 设置标签带驱动轴的电子齿轮比，使之开始加速

高速位置锁存(AL, _AL, RL)

Galil 运动控制器特有一路被称为高速地址锁存的专用数字输入信号，硬件上是一个边沿触发器。这个专用输入可以小于 0.1 μs 的分辨率识别输入的变化，并记录与之相对应的轴的位置。锁存输入在本应用中的作用是保持标签的登记位置准确。

当通过 **ALB** 命令启动锁存后，控制器在探测到输入状态变化的时刻会自动将编码器的位置值写入一个寄存器内。通过 **_RLB** 命令可以读取该寄存器。**_ALB** 返回一个布尔值用来指示位置锁存是否被启动。通过 **CN** 命令可以控制在上升沿 (**CN**, , 1) 或下降沿 (**CN**, , -1) 锁存位置。

计算

在本应用中，编写程序的第一个步骤就是以脉冲单位计算所需要的速度和距离。图 4 给出了机械原理图。纸箱以 $v=10\text{in/s}$ 的速度移动，长度为 10"，箱子之间的距离为 6"。标签长度为 10"，标签之间的间隙为 2"。登记标志设在标签带上每个标签间隙的中点处（距离每个标签的边沿 1"）。

由于箱子以一个固定的速度 v 移动，而标签带需要在时间 t 内加速到 v ，因此在给定的时间里箱子移动的距离 ($X_{\text{box}}=vt$) 是标签带移动距离 ($X_{\text{web}}=\frac{1}{2}at^2=\frac{1}{2}vt$) 的两倍。所以零位传感器被安装在距离贴标轮 2" ($2*1"$) 的位置上，因为贴标轮给标签留出了 1" 的启始头。标签带加速到 v 需要时间为 $t=2 X_{\text{web}}/v=(2*1\text{in})/(10\text{in/s})=0.2\text{s}$ ，而贴标的过程需要的时间为 $t=(10\text{in})/(10\text{in/s})=1\text{s}$ 。根据以上得到的数据，我们可以绘制出位置-时间图及速度-时间图（如图 5）。图中假设传送带已经达到了稳定的速度 10inches/second。




图 4. 贴标机在标签带开始加速瞬间的机械原理图




图 5. 贴标机和标签带的位置-时间及速度-时间图 (周期 1.6 秒)

根据图 4 中给出的轮子的直径及编码器的分辨率,表 2 中计算出了传送带及标签带驱动轮每英尺的线性分辨率脉冲数及每秒钟最大速度对应脉冲数。从表中可以得知,电子齿轮比 GR 应为: 3000 counts/s (驱动轮) : 4000 counts/s (传送带) = 0.75 : 1

表 2. 单位时间线性分辨率及速度脉冲数计算

轴	给定直径 (in)	周长 (in/rev)	给定编码器分辨率 (counts/rev)	线性分辨率 (counts/in)	要求速度 (in/s)	速度 (counts/s)
传送带	6.37	20.0	8000	400	10	4000
标签驱动轮	4.24	13.3	4000	300	10	3000

表 3. 以脉冲数为单位传送带的几个关键位置的计算

事件	时间(s)	传送带位置(in)	传送带位置(counts)
标签带开始加速 (图 4)	0	0	0
标签带达到指定速度, 开始进行贴标	0.2	2	800
贴标结束, 标签带开始减速	1.2	12	4800
标签带减速完成, 校验登记标志	1.4	14	5600
标签带静止状态	1.6	16	6400

程序描述

1. 带有#AUTO 的主程序在上电时自动执行。
2. 控制器控制传送带定位零点。将第一个箱子放在距离贴标轮 800 个脉冲的地方, 之后定义那里为零点。
3. 标签带驱动轴定位零点。零点定位在第一个标签的登记标志处。
4. 传送带开始旋转, 向贴标轮搬运箱子。假定每 16inch 一个箱子 (10inch 的箱子+6inch 的间隔), 那么程序就监测传送带的 6400, 12800, 19200 等位置。
5. 当箱子到达贴标的工作位置, 标签带驱动轮加速。在标签带加速的过程中, 传送带走过 800 个脉冲值的距离。
6. 一旦标签的速度到达了与箱子一致, 程序便开始等待传送带走过贴标过程需要的距离 :12" (4800 脉冲值)。
7. 当贴标完成之后, 标签带驱动轮在传送带的 800 个脉冲值的距离内, 脱离与传送带的电子齿轮啮合。
8. 当电子齿轮脱离完成后, 进行登记标志校验。
9. 如果发现登记标志, 电机会通过增量运动微调标签带以重新对齐登记标志。
10. 如果未发现登记标志, 则标签带驱动轮将带动标签带运动, 直至位置锁存被触发。一旦触发, 电机便通过增量运动微调标签带, 以重新对齐登记标志。
11. 重复上述过程

程序清单

以下给出这篇文章所用到的完整的程序，注释对程序进行了详细说明。

```
#AUTO ; '上电自动运行
' 为零点定位设置运动参数
AC*=500000 ; '设置加速度
DC*=500000 ; '设置减速度

' 对标签带卷轴使用开环偏压
KDC = 0 ; '设置微分常数为 0
KPC = 0 ; '设置比例常数为 0
KIC = 0 ; '设置积分常数为 0
OFC = -0.5 ; '设置一个负的偏压以保持标签带的张力。

' 高速定位传送带到零点传感器
SPA=100000 ; '设置速度
FEA ; '寻找零点开关
BGA ; '开始传送带运动
AMA ; '等待直到发现零点开关

' 缓慢逼近原点开关
SPA = 500 ; '为传动带设置一个低的速度
FEA ; '在低速状态下为传送带定位零点，以保证准确性
BGA ; '开始传送带运动
AMA ; '等待直到到达零点开关
DPA = 0 ; '将传送带位置设为 0

' 将标签带驱动轮的零点定位到标签带上的登记标志处
ALB ; '启动锁存检测第一个登记标志
JGB=250 ; '以 JOG 方式驱动标签带驱动轮朝登记标志运动
BGB ; '开始标签带驱动轮的运动
#Wait; JP#Wait, _ALB = 1 ; '等待直到发现登记标志
STB ; '停止标签带驱动轮
AMB ; '等待知道标签带驱动论运动完成
PAB = _RLB ; '将标签带驱动轮退回到登记标志所在位置
BGB ; '开始标签带驱动轮的运动
AMB ; '等待直到标签带驱动轮的运动结束
DPB=0 ; '定义此位置为 0

' 初始化电子齿轮
GAB = A ; '将传送带设置为电子齿轮的主动轴
GDB = 800 ; '标签带驱动轮在传送带的 800 个脉冲距离上与其啮合
```

```

'启动传送带
JGA = 4000          ; '为传送带轴设置 JOG 速度
BGA                ; '开始传送带的 JOG 运动

BoxPos = 0         ; '箱子的前沿位置： 0, 6400, 12800...

#Apply             ; '每个标签循环一次
ALB                ; '启动登记标志锁存

IF _TPX >= BoxPos ; '如果传送带的位置走到了一个新的箱子
GRB = 0.75         ; '设置标签驱动轮的电子齿轮比以启动加速
MF BoxPos + 4800   ; '等待直到贴标完成 (2" + 10")

GRB = 0            ; '开始以电子齿轮方式减速标签带驱动论
MF BoxPos + 5600   ; '等待直到标签带驱动轮减速完成(14")

BoxPos = BoxPos + 6400 ; '增加箱子脉冲位置(16")
Correct = 0        ; '登记传感器是否被触发

#Correct           ; '登记标志校验循环
SPB=100000        ; '设置校验速度

IF _ALB = 0        ; '如果登记传感器已被触发
PAB = _RLB         ; '退回到锁存的位置
BGB                ; '开始标签带驱动轮的运动
AMB                ; '等待直到标签带驱动轮的运动完成
Correct = 1        ; '指示可以退出#Correct 循环
ELSE                ; '如果登记传感器未被触发
IPB = 10           ; '标签带驱动轮增量运动 10 个脉冲值
AMB                ; '等待直到增量运动完成
ENDIF

JP#Correct, Correct = 0 ; '如果登记传感器未被触发则继续循环
ENDIF

JP#Apply           ; '下一个标签
EN

```